MCS 471 Project Four : Numerical Interpolation with Maple

The goal of this project is to investigate some numerical aspects of interpolation. In particular, we wish to understand the relation between the degree of the interpolating polynomial and the working precision. Also, we study the influence of the choice of the interpolation points on the accuracy of the result.

1. The Experiments in Maple

Below are the Maple commands you need to conduct the experiments. We use the function **PolynomialInterpolation** from the package **CurveFitting**.

```
> with(CurveFitting):
```

There are three parameters:

```
> Digits := 10:  # working precision
> n := 4:        # degree of interpolating polynomial
> b := 2:        # right bound on the interval [0,b]
```

We first generate our data sets, at equidistant, random, and Chebyshev points:

```
> h := b/n:
> x1 := [seq(evalf(i*h),i=0..n)];
> randomize():
> die := evalf(rand(0..10^Digits)/10^Digits):
> x2 := [seq(b*die(),i=0..n)];
> x3 := [seq(evalf(b/2*(1+cos((2*i+1)/(2*n+2)*Pi))),i=0..n)];
```

We evaluate the function e^x at the interpolation points:

```
> y1 := [seq(evalf(exp(x1[i+1])),i=0..n)];
> y2 := [seq(evalf(exp(x2[i+1])),i=0..n)];
> y3 := [seq(evalf(exp(x3[i+1])),i=0..n)];
```

Combining interpolation points and their function values gives us three data sets:

```
> data1 := zip((x,y) -> [x,y],x1,y1);
> data2 := zip((x,y) -> [x,y],x2,y2);
> data3 := zip((x,y) -> [x,y],x3,y3);
```

Then we have three polynomials, interpolating through these three data sets:

```
> p1 := PolynomialInterpolation(data1,z);
> p2 := PolynomialInterpolation(data2,z);
> p3 := PolynomialInterpolation(data3,z);
```

We make three error plots, one for each polynomial, and finally a combined error plot:

```
> plot(p1-exp(z),z=0..b);
> plot(p2-exp(z),z=0..b);
> plot(p3-exp(z),z=0..b);
> plot([p1-exp(z),p2-exp(z),p3-exp(z)],z=0..b);
```

This list of commands presented here show the set up for the numerical experiments. Instead of typing the commands, it is recommended that you download the worksheet from the web,
see http://www.math.uic.edu/~jan/MCS471.html.

2. The Assignments

In our experiment, we consider the interpolation of e^x over the interval $[0, b]$, for some bound b, with a polynomial of degree n. We investigate the following three items.

2.1 Influence of the precision, i.e: the parameter Digits. We say that the precision is *sufficiently high* when increasing the precision does not change the error plot.

For a given degree n, what should the value for Digits be to achieve a sufficiently high precision? Do experiments with $n = 5, 10, 15, 20$, for $b = 1$. For each n, find the value for Digits which reaches the sufficiently high precision. What happens to this value for Digits if we take $b = 10^2$ and $b = 10^6$?

Display the values for Digits in a table, with rows indexed by n, and three columns, for $b = 1, 10^2, 10^6$.

2.2 Choice of interpolation points. We distinguish three choices:

1. equidistant points in the interval;
2. randomly generated points;
3. Chebyshev points, defined as $\cos\left(\frac{2i+1}{2n+2}\pi\right)$, $i = 0, 1, \ldots, n$, for the interval $[-1, 1]$.

Compare the magnitude of the errors and the shape of the error plots for these three choices of interpolation points. In particular, for sufficiently high precision and $b = 1$ fixed, make the plots for increasing degrees $n = 5, 10, 15, 20$. As the degree goes up, what can you say about the shape of the plots and the convergence of the interpolating polynomial to the actual function e^x? Do the errors decrease as n increases for all three choices of interpolation points?

2.3 The error using Chebyshev points. The error of the interpolating polynomial contains the factor $(x - x_0)(x - x_1) \cdots (x - x_n)$. For $b = 1$, and $n = 5, 10, 15, 20$, do

```
> q := product(x-x3[k+1],k=0..n);
> plot(q,x=0..1);
```

What can you tell from the shape of the polynomial? What are the maxima and what are their values? Make a table with rows indexed by n and as columns the maximal value(s). Can you deduce a general relation between n and the maximal values?

3. The deadline is Friday 5 April, at 1PM

Bring your project solution to class. It should contain the following:

1. The answers to the first question consists of a table linking the value for Digits with the degrees $n = 5, 10, 15, 20$, and a statement about the general relation between the degree and the number of decimal places to reach a sufficiently high working precision.

2. For question two, you submit the plots of the errors. For each degree, write one line describing the comparison between the three choices of interpolation points. Finally, make a general statement about which choice is best.

3. The answer the third question contains again the plots with a line of explanation for each plot. Also state the general conclusion about the shape of the polynomial.

Additionally, but not instead of the answers above, you can hand in the print out of the Maple worksheet with your experiments. In this way you can still get some (partial) credit for your (partially) wrong answers. Write complete sentences without spelling and grammatical mistakes.

See http://www.math.uic.edu/~jan/MCS471.html for the hypertext version of this project. In particular, to avoid typing errors and to save time, you may want to download the Maple worksheet.

If you have questions, comments, or difficulties, feel free to come to my office for help.